

Tutorial para a Linguagem Julia

AUTOR

Vítor H. Nascimento

DATA DE PUBLICAÇÃO

14 de junho de 2023


Julia é uma linguagem criada recentemente, com o objetivo de criar código que rode aproximadamente tão rápido quanto em C, mas com a possibilidade de uso interativo e a facilidade de tratamento de estruturas complexas do *Matlab* ou do *Python*. Por ser *software* aberto, é uma alternativa muito interessante ao *Matlab* (que é caro) e ao *Octave* (que tende a ser muito lento).

Assim como *Matlab* e *Python*, *Julia* permite que você use objetos mais abstratos como vetores e matrizes de maneira bem natural, facilitando bastante a programação. No entanto, ao contrário de *Matlab* e *Python*, *Julia* consegue uma velocidade de processamento próxima da velocidade de C (desde que alguns cuidados sejam tomados).

Instalação

Para instalar *Julia*, há diversas opções. você pode baixar o programa diretamente da [Página principal da linguagem Julia](#), e usar um editor como o [Code](#) para criar um ambiente de desenvolvimento (IDE) no seu próprio computador.

Para instalar, é fácil: baixe o [arquivo executável](#) de acordo com o seu sistema operacional, e você já poderá programar. Com isso você já pode usar *Julia* através da linha de comando, e escrever programas usando um editor de texto qualquer. Vários editores de texto (como por exemplo *Emacs* e *Vim*) têm pacotes específicos para *Julia*. Uma boa opção é o [Code](#), que tem uma extensão bem completa para *Julia*, incluindo recursos para *debugging*. Para instalar a extensão, basta você entrar na aba *Extensions* na barra do lado esquerdo da janela principal do *Code*, e buscar o pacote *julia* (depois da instalação, veja na barra de status na parte de baixo da tela do *Code* que a instalação de *Julia* vai ser indexada, o que vai demorar um pouco. Ao terminar, você poderá abrir um terminal para usar os comandos de *Julia*, usando **CTRL-SHIFT-P** (**CMD-SHIFT-P** no Mac) e escolhendo **Julia: Start REPL**.



Code-InitScreen.png

Além do *Code*, você também pode usar *Julia* usando *notebooks* do [Jupyter](#) (para isso instale o pacote *IJulia*), ou *notebooks* do [Pluto](#). Para usar o *Jupyter*, você pode tanto usar uma instalação prévia do *Python* com os pacotes de *Jupyter* já instalados, ou aproveitar a opção de instalar tudo diretamente do *IJulia*, usando os comandos (dentro de uma janela do *Julia* - na listagem abaixo foi usado o comando `]` para entrar no modo de instalação de pacotes)

```
julia> ]  
pkg> add IJulia  
julia> using IJulia  
julia> notebook()
```

O programa vai então perguntar se você deseja instalar o *Jupyter*. Note que após a instalação, você não precisa mais rodar o comando `add IJulia`.

Também é possível usar uma instalação prévia do *Python* e do *Jupyter*. Se você já tiver o *Python* instalado no seu computador e quiser aproveitar, antes do comando `add IJulia` (e antes de entrar no modo de pacotes), dê o comando

```
ENV["PYTHON"] = "comando completo para chamar o Python"
```

Por exemplo, no meu computador o comando é (repare que comando para rodar o *Python* deve estar entre aspas)

```
julia> ENV["PYTHON"] = "/opt/local/bin/python3.10"
```

Antes de usar diversos pacotes é necessário instalá-los (alguns, como *LinearAlgebra*, são pré-instalados). Para fazer isso, entre no modo de pacotes digitando `]`. O cursor vai mudar de

```
julia>
```

para

```
(@v1.9) pkg>
```

No modo de pacotes, dê o comando `add Nome_do_pacote`, como

```
(@v1.9) pkg> add Plots
```

e o pacote desejado será baixado e instalado. Isso demora um pouco.

Também é bom atualizar os pacotes instalados de vez em quando, rodando (no modo de pacotes) o comando

```
(@v1.9) pkg> update
```

Para instalar o *Pluto*, use

```
julia> ]  
(@v1.9) pkg> add Pluto  
julia> using Pluto
```

Ajuda

A documentação geral da linguagem está [aqui](#). *Julia* também tem diversos pacotes de funções que podem ser instalados pelo usuário (veja como abaixo na seção [Gráficos](#)). Uma lista de pacotes com a documentação correspondente você encontra [neste endereço](#).

Na linha de comando, você pode obter ajuda para qualquer comando digitando `?Nome_do_comando`

Se você digitar `?` no início da linha de comando, *Julia* entra no modo de help, e o cursor muda para

```
help?>
```

digite o nome de um comando e virá uma mensagem de ajuda.

Também é possível rodar um comando do `shell` (caso você use Linux ou Mac), digitando `;` no início da linha de comando. O cursor muda para

```
shell>
```

Atenção, no entanto, que caracteres-curinga como `*` não funcionam nesse shell simples.

Comandos básicos e entrada de dados

Tanto a linha de comando em *Julia* (o [REPL \(read-eval-print loop\)](#)) quanto o *Jupyter* ou o *Pluto* permitem que você use a linguagem de maneira interativa. Este arquivo que você está lendo foi criado no *Jupyter*, assim como a maior parte dos exemplos das aulas. Você pode criar variáveis e

usá-las diretamente, como nos comandos a seguir:

```
n=2
```

2

```
x=2.5
```

2.5

No *Jupyter* e no *REPL*, se você terminar a linha com um ponto-e-vírgula, o resultado da operação não é mostrado. Isso é útil para definições, principalmente de vetores e matrizes (isso não é necessário dentro de uma função, nem em um *script*):

```
y=3.5;
```

Uma característica muito interessante de *Julia* é a facilidade de trabalhar com variáveis de diferentes tipos: inteiros, racionais, reais (ponto flutuante), complexos, vetores, matrizes, etc. Em muitas situações você pode trabalhar com esses diferentes tipos de variáveis como faria com papel e lápis (o que também pode ser feito em *Matlab* e *Python*), e *Julia* se encarrega de transformar os formatos como necessário para você (na maior parte das vezes, veremos algumas exceções e o motivo para elas mais à frente).

As quatro operações normais são obtidas com os operadores `+`, `-`, `*` e `/`. Exponenciação é obtida com o operador `^`.

```
z=n*x+n
```

7.0

Em *Julia* podemos definir variáveis inteiras, escrevendo números sem ponto decimal, como

```
m=7
```

7

```
typeof(m)
```

Int64

Podemos definir variáveis reais (ponto flutuante), escrevendo o ponto decimal (seguido ou não de zero), como

```
z=7.0
```

7.0

```
typeof(z)
```

Float64

As quatro operações usam os comandos usuais, por exemplo

```
8*n+11
```

27

Veja que, como todos os operandos são números inteiros, o resultado também é um número inteiro (repare que a saída 27 não tem ponto).

Em *Julia* não é necessário escrever explicitamente o `*` ao se multiplicar uma constante por uma variável como em `8n`.

```
8n
```

16

```
2.5n
```

5.0

Isso só vale no caso de se colocar um número constante na frente de uma variável, como `8n`, `2.5x`, etc. O símbolo `n8` representa uma variável diferente:

```
n8=11
```

11

```
8n+n8
```

27

Outra característica divertida de *Julia* é que podemos definir variáveis com letras gregas:

```
μ=0.5
```

0.5

```
2μ
```

1.0

As letras gregas podem ser obtidas digitando-se o nome da letra (em

inglês), precedido por `\` (são os códigos usados em LaTeX) e seguido por `TAB`: `\alpha[TAB]`, `\beta[TAB]`, `\gamma[TAB]`. Letras maiúsculas gregas são obtidas como `\Gamma[TAB]`, por exemplo.

`$\alpha=1.5$`

1.5

`$\beta=0.7$`

0.7

`$\Gamma=2/3$`

0.6666666666666666

`$\alpha+\Gamma*\beta$`

1.9666666666666666

O símbolos `π` e `pi` são pré-definidos, e representam a mesma coisa:

`π`

$\pi = 3.1415926535897\dots$

`pi`

$\pi = 3.1415926535897\dots$

`$\cos(\pi/3+2\pi)$`

0.5000000000000006

O resultado da última operação é guardado na variável `ans`:

`$8n+11$`

27

`$ans/3$`

9.0

Repare que o resultado de uma divisão usando `/` é um número em ponto flutuante, mesmo quando o dividendo e o divisor são inteiros e o resultado poderia ser representado exatamente como um inteiro. Para fazer divisão

entre inteiros, você pode usar o operador \div (que você obtém escrevendo `\div` seguido por `TAB`).

```
(8n+11)÷3
```

9

Repare que, se o primeiro número não for divisível pelo segundo, \div fornece a parte inteira da divisão:

```
8÷3
```

2

Por outro lado, $/$ fornece o resultado em ponto flutuante:

```
8/3
```

2.6666666666666665

O resto da divisão inteira é obtido com a função `mod` ou o operador `%`:

```
8 % 3
```

2

```
mod(8,3)
```

2

Continuação de linhas de comando

Se a linha de comando que você for digitar for muito longa, você pode continuar na próxima linha sem maiores problemas. Basta você apertar `[ENTER]` apenas num ponto em que ficar óbvio para o *parser* que o comando ainda não terminou (por exemplo, no meio de um parênteses, ou logo após um símbolo de operação matemática com dois argumentos, como `+`, `-`, `/`, `*`):

```
x = 2.5cos(π / 5) *  
    exp(-2)
```

0.27372136019241655

Neste caso, como a primeira linha foi terminada com um `*`, o *parser* espera que a próxima linha contenha a continuação do comando. Se você parar a primeira linha imediatamente antes do `*`, no entanto, você teria parado a primeira linha em um comando completo, e o *parser* iria interpretar a segunda linha como um erro:

```
x = 2.5cos(π / 5)
    * exp(-2)
```

LoadError: syntax: "*" is not a unary operator

Cuidado no entanto com $+$ e $-$: esses símbolos têm dois significados: operação de soma e subtração (uma operação que necessita de dois argumentos), ou especificação do sinal de um número (uma operação que precisa de um único argumento, o que a mensagem de erro acima chamou de `unary operator`). Veja:

```
x = 2.5cos(π / 5)
    - exp(-2)
```

-0.1353352832366127

Agora a saída foi $-e^{-2}$, sem erro, e o símbolo `x` contém o valor de $2.5 \cos(\pi/5)$. No primeiro exemplo acima o valor de `x` ficou igual a $2.5 \cos(\pi/5)e^{-2}$. Compare os dois trechos de código abaixo:

```
x = 1
    + 2
```

2

e

```
y = 1 +
    2
```

3

Vamos comparar os valores de `x` e `y`:

```
println("x = $x")
println("y = $y")
```

x = 1

y = 3

Aproveite para reparar que um cifrão dentro de uma `string` imprime o resultado da operação logo depois do cifrão, assim, por exemplo `"$x"` é uma `string` com o valor da variável `x`. Se você quiser uma `string` com `$x`, use `\` antes do cifrão: `"\$x"`

```
println("10x+y = $(10x+y)")
println("\$x")
```

10x+y = 13

\$x

Números racionais

Além de números inteiros e reais, podemos trabalhar também com frações (números racionais) diretamente em *Julia*, usando `//` em vez de `/`:

```
q=8//3
```

8//3

Aqui não foi feita nenhuma conta - a variável q representa uma fração, e podemos fazer operações com frações normalmente. Por exemplo,

```
q+5//6
```

7//2

Veja que o resultado já vem simplificado. Podemos usar o resultado como um número qualquer:

```
ans*n
```

7//1

Repare que o resultado continua sendo uma fração. Na verdade, ao fazer uma conta com números de tipos diferentes, *Julia* "promove" os tipos mais restritivos para o tipo mais geral. Por exemplo, como os racionais contêm os inteiros, a soma de um racional com um inteiro resulta em um racional:

```
ans+1
```

8//1

Já a soma de um racional (ou de um inteiro) com um número em ponto flutuante resulta um número em ponto flutuante:

```
ans+1.0
```

9.0

Obviamente, não faz sentido escrever um número racional em que o numerador ou o denominador não seja inteiro (quer dizer, com numerador ou denominador em ponto flutuante) - se você fizer isso, você recebe uma mensagem de erro:

```
2.0//3
```

```
LoadError: MethodError: no method matching //(::Float64,  
::Int64)
```

```

Closest candidates are:
 (::Union{Number, AbstractPolynomial}, PQ::Polynomials.AbstractRationalFunction)
 where PQ<:Polynomials.AbstractRationalFunction
 @ Polynomials Polynomials
 ~/.julia/packages/Polynomials/fboHv/src/rational-
 functions/common.jl:120:
 (::AbstractPolynomial, ::Number)
 @ Polynomials Polynomials
 ~/.julia/packages/Polynomials/fboHv/src/rational-
 functions/common.jl:129:
 (::Integer, ::Integer)
 @ Base Base
 4rational.jl:62:
 ...

```

Vamos explicar melhor como entender as mensagens de erro em outro capítulo mais à frente.

Números complexos

Também podemos definir números complexos usando o número pré-definido `im` (a unidade imaginária):

```
z=1+2im
```

```
1 + 2im
```

```
z^2
```

```
-3 + 4im
```

```
im^2
```

```
-1 + 0im
```

Você pode fazer qualquer operação habitual com números complexos. Por exemplo, o módulo do número z é

```
abs(z)
```

```
2.23606797749979
```

Lembrando que

$$e^{j\pi/3} = \cos(\pi/3) + j \operatorname{sen}(\pi/3) :$$

```
z1=exp((pi/3)im)
```

```
0.5000000000000001 + 0.8660254037844386im
```

Também podemos obter as partes real e imaginária, e o ângulo (em

radianos) de um número complexo com os comandos

```
real(z1)
```

```
0.5000000000000001
```

```
imag(z1)
```

```
0.8660254037844386
```

```
angle(z1)
```

```
1.0471975511965976
```

Podemos converter o ângulo para graus facilmente:

```
ans*180/π
```

```
59.99999999999999
```

Vetores e matrizes

A grande inovação feita pelo *Matlab* foi a capacidade de trabalhar com vetores e matrizes de maneira natural, sem a necessidade de se escrever explicitamente laços e subrotinas para toda operação. Essa ideia foi usada também em *Python* na biblioteca *NumPy*, e em *Julia*. No caso de *Julia* podemos definir vetores e matrizes com uma sintaxe semelhante à do *Matlab*, usando `[` e `]` para marcar o início e o final do vetor ou matriz, e `;` para definir uma nova linha. Assim, por exemplo

```
a=[1 2 3;4 5 6]
```

```
2×3 Matrix{Int64}:
```

```
1 2 3  
4 5 6
```

resulta uma matriz de duas linhas e três colunas com elementos inteiros. Podemos acessar os elementos de `a` como a seguir:

```
a[1,2]
```

```
2
```

```
a[2, 3]
```

```
6
```

Preste atenção no fato que o primeiro elemento de cada linha ou coluna é

indexado pelo número 1 (ao contrário de *C* e *Python*, em que o primeiro elemento é indexado por 0). Assim, o primeiro elemento de `a` é dado por `a[1,1]`:

```
a[1,1]
```

1

Note que o pacote `OffsetArrays` pode ser usado em *Julia* para definir vetores e matrizes indexados como desejado. Isso será discutido em mais detalhes mais à frente.

Também é importante prestar atenção que *Julia* usa colchetes para acessar os elementos de um vetor ou matriz, enquanto que *Matlab* usa parêntesis: o comando anterior em *Matlab* seria `a(1,1)`.

Outra inovação muito interessante e útil feita pelo *Matlab*, e que *Julia* também usa, é a notação ":" para escolher pedaços de matrizes e vetores. Para ver como essa ideia funciona, considere a matriz com números aleatórios

```
b=randn(3,5)
```

3×5 Matrix{Float64}:

```
-1.20257  0.0275332  0.344687  1.35141  -0.388319  
-0.708781 -0.118693 -0.706623  1.34241  1.12536  
-1.1721   1.82418   1.41118   0.772849  1.6824
```

Se quisermos tomar o bloco composto pelos elementos

$$\begin{bmatrix} b[1,2] & b[1,3] & b[1,4] \\ b[2,2] & b[2,3] & b[2,4] \end{bmatrix},$$

podemos usar o comando

```
b[1:2,2:4]
```

2×3 Matrix{Float64}:

```
0.0275332  0.344687  1.35141  
-0.118693 -0.706623  1.34241
```

O operador ":" permite criar sequências de números. Assim, "2:4" é uma lista com os números 2, 3, 4

```
n=2:4
```

2:4

```
n[1]
```

2

```
n[2]
```

3

```
n[3]
```

4

Há uma outra forma do operador `:`, em que se acrescenta o passo da sequência - `1:2:5` é a lista com os números 1, 3, 5

```
n1=1:2:5
```

1:2:5

```
n1[1]
```

1

```
n1[2]
```

3

```
n1[3]
```

5

```
b[1:2:3,n1]
```

```
2x3 Matrix{Float64}:  
-1.20257  0.344687 -0.388319  
-1.1721   1.41118   1.6824
```

Para pegar uma linha ou uma coluna inteira de uma matriz, você pode usar `:` sozinho, como nos exemplos abaixo

```
b[:,1] # Primeira coluna de b
```

```
3-element Vector{Float64}:  
-1.2025718956626734  
-0.7087805958367919  
-1.172099344274593
```

```
b[2,:] # Segunda linha de b
```

```
5-element Vector{Float64}:  
-0.7087805958367919
```

```
-0.11869321934393834
-0.7066226500613605
 1.342406318908508
 1.1253639949151348
```

Note que, como o formato-padrão para vetores em *Julia* é de vetores-coluna, `b[2,:]` resulta em um vetor-coluna. Se você precisa da segunda linha de `b` em formato de vetor-linha, use `b[[2],:]`.

```
b[[2],:]
```

```
1×5 Matrix{Float64}:
 -0.708781 -0.118693 -0.706623  1.34241  1.12536
```

O resultado é diferente porque `2` é um número inteiro, enquanto que `[2]` é um vetor com um único elemento.

O passo usado com o operador `:` pode ser um número qualquer, inclusive negativo. Por exemplo,

```
x=0:0.1:2
```

```
0.0:0.1:2.0
```

```
x[1]
```

```
0.0
```

```
x[2]
```

```
0.1
```

```
x[3]
```

```
0.2
```

```
ninv=5:-1:1
```

```
5:-1:1
```

```
ninv[1]
```

```
5
```

```
ninv[2]
```

```
4
```

A variável `end` pode ser usada apenas ao se acessar os elementos de um vetor ou matriz, e representa o último elemento da dimensão correspondente (analogamente, `begin` representa o primeiro elemento):

```
b
```

```
3x5 Matrix{Float64}:  
-1.20257  0.0275332  0.344687  1.35141  -0.388319  
-0.708781 -0.118693 -0.706623  1.34241  1.12536  
-1.1721   1.82418   1.41118   0.772849  1.6824
```

```
b[1,end]
```

```
-0.38831860340214563
```

```
b[end,begin]
```

```
-1.172099344274593
```

```
b[end,end]
```

```
1.6824027477984256
```

Podemos inverter a ordem dos elementos da matriz usando `end:-1:1` (ou `end:-1:begin`):

```
b[1:2,end:-1:begin]
```

```
2x5 Matrix{Float64}:  
-0.388319  1.35141  0.344687  0.0275332  -1.20257  
 1.12536   1.34241 -0.706623 -0.118693  -0.708781
```

```
b[end:-1:1,end:-1:1]
```

```
3x5 Matrix{Float64}:  
 1.6824  0.772849  1.41118  1.82418  -1.1721  
 1.12536  1.34241 -0.706623 -0.118693 -0.708781  
-0.388319  1.35141  0.344687  0.0275332  -1.20257
```

Operações com matrizes podem ser feitas usando os operadores aritméticos normalmente:

```
a
```

```
2x3 Matrix{Int64}:  
 1  2  3  
 4  5  6
```

```
b=[1 1;2 2]
```

```
2x2 Matrix{Int64}:
```

```
1 1  
2 2
```

```
b*a
```

```
2x3 Matrix{Int64}:
```

```
5 7 9  
10 14 18
```

Como acontece com vetores e matrizes, para o produto estar bem definido, o número de colunas da primeira parcela deve ser igual ao número de linhas da segunda parcela, senão o produto não é definido, e você recebe uma mensagem de erro:

```
a*b
```

```
LoadError: DimensionMismatch: matrix A has dimensions (2,3),  
matrix B has dimensions (2,2)
```

É possível multiplicar escalares por matrizes e somar ou subtrair matrizes de mesma dimensão normalmente:

```
c=[1 2;3 4]
```

```
2x2 Matrix{Int64}:
```

```
1 2  
3 4
```

```
d=2b+c
```

```
2x2 Matrix{Int64}:
```

```
3 4  
7 8
```

Também pode-se multiplicar os elementos de duas matrizes elemento-a-elemento, usando `.*` em vez de `*`. Compare:

```
b*c
```

```
2x2 Matrix{Int64}:
```

```
4 6  
8 12
```

é um produto normal de matrizes. Por outro lado, o produto elemento-a-elemento resulta

```
b.*c
```

2x2 Matrix{Int64}:

```
1 2
6 8
```

Também é possível usar o operador de exponenciação com matrizes quadradas:

```
b^2
```

2x2 Matrix{Int64}:

```
3 3
6 6
```

Repare que o operador `.^` representa também a exponenciação termo-a-termo:

```
b.^2
```

2x2 Matrix{Int64}:

```
1 1
4 4
```

Vetores são definidos de modo similar:

```
v=[1,2]
```

2-element Vector{Int64}:

```
1
2
```

Note que você pode usar vírgula ou ponto-e-vírgula para definir um vetor:

```
v1=[2;3]
```

2-element Vector{Int64}:

```
2
3
```

No entanto, a vírgula e o ponto-e-vírgula têm na verdade um significado um pouco diferente: ponto-e-vírgula é o operador de concatenação, assim, você pode obter um vetor de quatro elementos fazendo

```
v4 = [v; v1]
```

4-element Vector{Int64}:

```
1
2
2
3
```

Isto é equivalente à função `vcat`:

```
v4alt = vcat(v, v1)
```

4-element Vector{Int64}:

1
2
2
3

Se você usar vírgula, vai ter uma surpresa:

```
vv = [v, v1]
```

2-element Vector{Vector{Int64}}:

[1, 2]
[2, 3]

Você obteve um *vetor de vetores*! Repare:

```
vv[1]
```

2-element Vector{Int64}:

1
2

```
vv[2]+vv[1]
```

2-element Vector{Int64}:

3
5

A vírgula separa elementos de um vetor, o ponto-e-vírgula serve para concatenar vetores (ou matrizes). No caso dos elementos serem escalares (números), usar os dois dá o mesmo resultado, mas em geral o resultado é diferente, como acabamos de ver. Veja que no caso de um vetor de vetores, os elementos individuais de cada vetor podem ser acessados assim:

```
vv[1][2]
```

2

Voltando a vetores e matrizes comuns, o produto de uma matriz por um vetor é obtido como usualmente:

```
b*v
```

2-element Vector{Int64}:

3
6

Da maneira como definido acima, v é um vetor-coluna. Para definir um vetor-linha, podemos usar espaços entre os elementos:

```
u=[1 2]
```

```
1×2 Matrix{Int64}:  
 1 2
```

Como `u` é um vetor linha, pode multiplicar `b` pela esquerda:

```
u*b
```

```
1×2 Matrix{Int64}:  
 5 5
```

Usado em matrizes ou vetores, o operador `'` realiza a operação de transposição:

```
b'
```

```
2×2 adjoint(::Matrix{Int64}) with eltype Int64:  
 1 2  
 1 2
```

Veja que `'` também pode ser usado para criar um vetor linha:

```
v'
```

```
1×2 adjoint(::Vector{Int64}) with eltype Int64:  
 1 2
```

O tipo do resultado (`adjoint`) é diferente por detalhes de implementação que não são importantes neste momento.

O operador `:` gera uma variável do tipo `Range`

```
u=1:2
```

```
1:2
```

```
typeof(u)
```

```
UnitRange{Int64}
```

Essas variáveis quase sempre funcionam como se fossem um vetor-coluna. A grande diferença é que o vetor não é criado verdadeiramente, apenas fica indicado para ser calculado quando necessário (o que permite ganhos de economia de memória e eficiência). Para converter para um vetor comum, use

```
uvec = collect(u)
```

```
2-element Vector{Int64}:  
 1  
 2
```

No entanto, em quase todas as situações os `Range` podem ser usados

como se fossem um vetor comum:

```
b*u
```

```
2-element Vector{Int64}:
```

```
3
```

```
6
```

Multiplicar b pela esquerda agora gera um erro por incompatibilidade de dimensões:

```
u*b
```

```
LoadError: DimensionMismatch: matrix A has dimensions (2,1),  
matrix B has dimensions (2,2)
```

O número de elementos em um vetor ou `Range` pode ser obtido com o comando `length`. As dimensões de uma matriz podem ser obtidas com o comando `size`:

```
length(u)
```

```
2
```

```
size(a)
```

```
(2, 3)
```

A inversa de uma matriz é obtida com a função `inv`

```
inv(c)
```

```
2×2 Matrix{Float64}:
```

```
-2.0  1.0
```

```
1.5  -0.5
```

```
c*inv(c)
```

```
2×2 Matrix{Float64}:
```

```
1.0      0.0
```

```
8.88178e-16  1.0
```

Para resolver o sistema de equações

$$cx = v$$

é possível usar

$$x = \text{inv}(c) * v,$$

mas isso é pouco eficiente (usa um número exageradamente grande de operações). É melhor usar o algoritmo de eliminação de Gauss com o

operador `\`:

```
x=c\v
```

2-element Vector{Float64}:

```
0.0  
0.5
```

Exceto por erros numéricos, o resultado deve ser igual ao de

```
inv(c)*v
```

2-element Vector{Float64}:

```
0.0  
0.5
```

O operador `\` assume que `x` e `v` são vetores-coluna. Caso sejam vetores-linha, e a inversa deva ser multiplicada pela direita, você pode usar `/`:

```
v'/c
```

1×2 adjoint(::Vector{Float64}) with eltype Float64:

```
1.0 0.0
```

Isto é equivalente a resolver o sistema

$$c'y = v,$$

como podemos ver com o comando

```
y=c'\v
```

2-element Vector{Float64}:

```
1.0  
0.0
```

Para calcular o produto escalar entre dois vetores, você pode usar duas formas. Uma é simplesmente usar a transposta, e fazer

```
v'*v
```

5

A outra maneira é usar o operador `\cdot` (obtido digitando-se `\cdot[TAB]`), disponível na biblioteca de Álgebra Linear.

Para carregar uma biblioteca, use o comando `using`:

```
using LinearAlgebra
```

```
v·v
```

Veja que agora não foi necessário transpor o vetor v .

Aplicação de funções a vetores ou matrizes

Uma outra ideia muito útil do *Matlab* que foi usada em *Python* e *Julia* é a possibilidade de aplicar funções quaisquer a todos os elementos de um vetor. Por exemplo, considere a função `cos()`:

```
cos(pi/3)
```

```
0.5000000000000001
```

Vamos calcular a função $\cos(x)$ para os valores $x = 0, \pi/6, \pi/3, \pi/2, 2\pi/3, 5\pi/6, \pi$:

```
x=0:pi/6:pi
```

```
0.0:0.5235987755982988:3.141592653589793
```

```
y=cos.(x)
```

```
7-element Vector{Float64}:
```

```
 1.0
 0.8660254037844387
 0.5000000000000001
 6.123233995736766e-17
 -0.4999999999999998
 -0.8660254037844385
 -1.0
```

Repare que escrevemos `cos.(x)`, não `cos(x)`. Qualquer função que aceite um argumento escalar pode ser aplicada a todos os elementos de um vetor (ou matriz) desde que seja colocado um `.` antes do parêntesis:

```
x=0:3
```

```
0:3
```

```
exp.(-x)
```

```
4-element Vector{Float64}:
```

```
 1.0
 0.36787944117144233
 0.1353352832366127
 0.049787068367863944
```


Gráficos

Há diversas bibliotecas para desenhar gráficos em *Julia*. Uma muito interessante de usar é a biblioteca [Plots](#), que define uma interface comum que pode ser usada com diversos geradores de gráficos (*backends*). Algumas opções são [GR](#), [PythonPlot](#), [PlotlyJS](#). [GR](#) é instalado juntamente com [Plots](#), os outros precisam ser instalados separadamente.

A biblioteca de funções para desenhar gráficos do *Matlab* é tão boa que foi basicamente clonada em *Python* (a biblioteca *Matplotlib*). E, como *Julia* pode facilmente usar código em *Python*, uma das bibliotecas para gráficos em *Julia* simplesmente usa o pacote do *Python*, através da biblioteca [PythonPlot](#) (Em computadores mais antigos é necessário usar a biblioteca [PyPlot](#) em vez da [PythonPlot](#). O uso das duas é semelhante).

No restante deste documento vamos exemplificar os comandos gráficos usando *Plots* e o backend default (*GR*). Ao longo do curso também serão apresentados diversos exemplos usando o pacote *Plots* com o backend *PythonPlot* e com o backend *PlotlyJS*, que tem a vantagem de permitir gráficos interativos no *Jupyter* (O *PythonPlot* também gera gráficos interativos, mas não quando se está usando o *Jupyter*).

Caso o pacote [Plots](#) ainda não esteja instalado, instale-o com o comando

```
]add Plots
```

Depois da instalação, volte para o modo normal do *REPL* e carregue o pacote para uso com o comando


```
using Plots
```

Vamos terminar os comandos a seguir com ponto-e-vírgula para suprimir a impressão da saída.

```
x=0:π/100:2π;
```

```
y=cos.(5x) .* exp.(-0.5x);
```


```
plot(x,y)
```



O gráfico pode ser melhorado com diversos comandos: `xlabel!()` acrescenta uma legenda no eixo x , `ylabel!()` acrescenta uma legenda no eixo y , `title!()` acrescenta um título:

```
plot(x,y)
xlabel!("x")
ylabel!("y")
title!("Função y(x)")
```


Função $y(x)$



Você também pode usar comandos do LaTeX para incrementar as legendas, colocando um `L` antes da *string*, e acrescentar legendas as gráficos usando o comando `label` dentro do comando `plot`, como abaixo:

```
using LaTeXStrings
plot(x,y, label = L"y(x)")
xlabel!(L"x")
ylabel!(L"y")
title!(L"y(x)=\cos(5x)e^{-\frac{x}{2}}")
```


$$y(x) = \cos(5x)e^{-\frac{x}{2}}$$



```
plot(x,y, label = L"y(x)")
xlabel!(L"x")
ylabel!(L"y")
title!(L"Também posso misturar texto e equações:
        $y(x)=\cos(5x)e^{-\frac{x}{2}}$")
```


Também posso misturar texto e eq

$$y(x) = \cos(5x)e^{-\frac{x}{2}}$$




Se você quiser desabilitar a legenda, use `label = ""`.

```
plot(x,y, label="")
```




Como no *Matlab*, há funções para gerar gráficos em escala logarítmica:

```
x1=10.^(-2:0.1:2)
y1=exp.(x1)
plot(x1,y1, yaxis = :log)
xlabel!("x")
ylabel!("y")
```




Também é possível gerar gráficos com o eixo x em escala logarítmica usando o comando `plot(x,y, xaxis = :log)`, e com as duas escalas logarítmicas, usando o comando `plot(x,y, xaxis = :log, yaxis = :log)`.

```
x2 = 0.01:0.001:5
y2=1 ./x2
plot(x2, y2, label = "", xaxis = :log,
     xlabel = L"x", ylabel = L"y")
```



```
plot(x2, y2, xlabel = L"x", ylabel = L"y",  
      xaxis = :log, yaxis = :log, label = "")
```




Você talvez tenha reparado que os comandos `xlabel`, `ylabel`, `title`, etc, podem ser fornecidos como parâmetros do comando `plot`, como em

`plot(x, y, xlabel = L"x")`, ou em comandos separados, como `plot(x,y); xlabel(L"x")`. No pacote `Plots`, toda função terminada com um ponto de exclamação (`xlabel!`, `title!`, `plot!`) significa que o comando modifica um comando `plot` imediatamente anterior. Assim, se você quiser acrescentar uma segunda função a um gráfico começado anteriormente, você pode usar `plot!`, como no exemplo a seguir:

```
plot(x, y, label = L"y(x)", xlabel = L"x")
plot!(x2, y2/100, label = L"\frac{y_2(x)}{100}",
       title = "Duas funções na mesma figura")
```


Duas funções na mesma figura



Também é possível combinar dois traços em um só gráfico usando uma matriz para definir as ordenadas, em que cada coluna representa um traço diferente, como abaixo:

```
y = [y sin.(10x).*exp.(-(x.-3).^2)]
plot(x, y, label = [L"\cos(5x)e^{-x/2}" L"\sin(10x) e^{-(x-3)^2}"]
      xlabel = L"x", title = "Duas funções, forma alternativa")
```

Duas funções, forma alternati



Preste atenção que o argumento de `label` **não** deve ter vírgula entre cada legenda: o comando é `label = ["leg.1" "leg.2"]`, não `label = ["leg.1", "leg.2"]` (você usa `L"leg"` se for usar comandos do LaTeX na legenda, caso contrário o `L` inicial não é necessário).

Condições e laços

Condições em *Julia* seguem o padrão if-then-else habitual em várias linguagens, como mostra o exemplo abaixo:

```
a=1.0
if a<1
    print("menor")
elseif a==1
    print("igual")
else
    print("maior")
end
```

igual

```
if a!=1
    print("diferente")
else
    print("igual")
end
```

igual

Repare que testes são feitos sempre sobre variáveis do tipo `Bool`. Por exemplo,

```
a=true
if a
    print("Verdadeiro")
else
    print("Falso")
end
```

Verdadeiro

É possível encadear testes em uma única linha:

```
a = 3
if 1 < a < 4
    println("a ∈ [1, 4]")
end
if 1 < a != 3
    println("a > 1 e a ≠ 3")
else
    println("a ≤ 1 ou a = 3")
end
```

$a \in [1, 4]$

$a \leq 1$ ou $a = 3$

```
a = 2
if 1 < a < 4
    println("a ∈ [1, 4]")
end
if 1 < a != 3
    println("a > 1 e a ≠ 3")
else
    println("a ≤ 1 ou a = 3")
end
```

$a \in [1, 4]$

$a > 1$ e $a \neq 3$

Ao contrário do *Matlab*, testes não podem ser feitos diretamente sobre variáveis numéricas só sobre variáveis do tipo `Bool`, que assumem os valores `true` ou `false`. Ou seja, o código

```
x = 1
if x
    Do something
end
```

funcionaria em *Matlab*, mas não funciona em *Julia*. Em *Julia*, você tem de usar `if x == 1`.

Laços `for` e `while`

Laços em *Julia* podem ser criados com os comandos `for` e `while`, como nos exemplos a seguir:

```
soma=0
for i=1:10
    soma += i
end
print(soma)
```

55

É possível escrever `for i in 1:10` ou `for i ∈ 1:10` (`∈` é obtido com `\in[TAB]`).

```
soma=0; i=0
while i<=10
    soma += i
    i+=1
end
print(soma)
```

55

Também é possível definir laços sobre vetores:

```
a=[1 3 2 7 -8]
soma=0
for x in a
    soma += x^2
end
print(soma)
```

127

Uma outra alternativa, que é útil para variáveis mais complexas é a função `eachindex`, que retorna um `iterator`, um objeto que permite acessar todos os elementos do vetor ou matriz na ordem mais eficiente:

```
soma = 0
for i in eachindex(a)
    soma += a[i]^2
end
```

Repare apenas que em *Julia* há formas de se calcular os resultados dos exemplos assim de maneira mais compacta - note que a maneira compacta não é necessariamente a mais rápida em *Julia*, mas torna o código mais simples de ler e escrever:

```
sum(a.^2)
```

127

Laços em *scripts* e em funções

Uma advertência: se você estiver usando *Julia* a partir de um arquivo *script* (quer dizer, um conjunto de comandos colocados em um arquivo de texto, que é executado usando o comando `include` ou a opção para executar um arquivo do *VS Code*). Em um laço dentro de um *script* que estiver no ambiente global (isto é, fora de uma função), o programa não enxerga variáveis externas. Isto acontece apenas em *scripts* em que o código que está sendo executado está fora de uma função. Assim, nos exemplos anteriores, a variável *soma* não estaria disponível dentro dos laços. Isto não acontece com comandos digitados diretamente no *REPL*, que é como os exemplos mostrados aqui estão sendo executados. O motivo para essa diferença é meio complicado de explicar, diz respeito a implementações de variáveis globais.

Há duas soluções para isto: Se o laço `for` ou `while` estiver definido dentro de uma função ou diretamente na linha de comando do *REPL*, ou no ambiente do *Jupyter*, variáveis externas estarão disponíveis para o laço.

Se você estiver num *script*, e definir um laço diretamente sem criar uma função, é necessário modificar um pouco o código colocando a diretiva `global soma` dentro do laço:

```
soma=0
for i=1:10
    global soma
    soma+=i
end
```

Isto ocorre porque todas as variáveis criadas fora de funções e laços são do tipo `global`, e laços herdam do ambiente em que são definidos as variáveis locais, mas não as variáveis globais.

Isso é um pouco confuso no início, mas a regra geral é a seguinte: se você criar um laço, e ao tentar usar uma variável definida fora dele, receber uma mensagem de erro do tipo `ERROR: UndefVarError: soma not defined`, basta acrescentar o comando `global soma` no início do laço.

Funções

Você pode criar funções em *Julia* em qualquer lugar - no meio do código da linha de comando, ou num arquivo em separado.

Se você criar a função em um arquivo em separado, deve carregá-la antes de usá-la com a função `include("Nome_do_arquivo.jl")`. Arquivos em *Julia* normalmente levam a extensão `.jl`.

Vamos criar uma função simples:

```
function cosexp(x)
    y=cos(5x)*exp(-0.5x)
    return y
end
```

cosexp (generic function with 1 method)


Uma vez definida, você pode usar a função normalmente:

```
cosexp(2)
```

-0.30867716521951294

Como a função `cosexp()` tem apenas um argumento escalar, você pode aplicá-la a todos os elementos de um vetor usando `cosexp.()` :

```
y1=cosexp.(x)
plot(x,y[:,1], label="y calculado em linha de comando",
     lw=5) # Desenha com linha mais grossa
plot!(x,y1, lw = 2,
      label="y calculado com a função cosexp()")
```



Repare que a saída da função é informada pelo operador `return`. Uma função pode ter mais de uma saída também, como por exemplo

```
function fasor(V)
    r=abs(V)
    θ=angle(V)*180/π
```

```
    return r,θ
end
```

fasor (generic function with 1 method)

As duas saídas são obtidas chamando-se a função como a seguir.

```
p, φ = fasor(1+im)
```

(1.4142135623730951, 45.0)

```
p
```

1.4142135623730951

```
φ
```

45.0

Você também pode definir funções simples na linha de comando:

```
ce = x -> cos(5x) * exp(-0.5x)
ce.(2.5) == cosexp(2.5)
```

true

Variáveis aleatórias

Você pode gerar números (pseudo-)aleatórios aproximando uma distribuição gaussiana padrão usando o comando `randn()`, e números aproximando uma distribuição uniforme no intervalo $[0, 1]$ usando o comando `rand()`:

```
randn()
```

-0.03329582951789793

```
randn()
```

-0.9106685314887488

```
randn()
```

1.178342088401758

```
randn()
```

```
-1.18941250070801
```

Também é possível gerar um vetor de números aleatórios:

```
x=randn(5)
```

```
5-element Vector{Float64}:
```

```
-0.22358537589199912  
 1.218283443090666  
-0.9842843113785062  
-1.14174813258054  
 0.5301344589522416
```

```
y=rand(5)
```

```
5-element Vector{Float64}:
```

```
0.025848076321400515  
0.28462698978329126  
0.18117338868816701  
0.568857094213763  
0.5749916042039813
```

Note que no *Matlab*, `rand(5)` gera uma matriz 5x5, e `rand(5,1)` gera um vetor. Em *Julia*, a matriz é criada com `rand(5,5)`.

Finalmente, pode-se criar matrizes aleatórias também:

```
R=randn(3,4)
```

```
3×4 Matrix{Float64}:
```

```
 1.87821  0.146566 -1.02526  -0.148178  
-0.698936  2.56779 -0.582682  0.234254  
 0.262441  0.470171 -0.0684889 -0.238154
```

O pacote `Distributions` permite que você calcule amostras e propriedades de diversas distribuições comuns.

```
using Distributions
```

Agora você pode criar “variáveis aleatórias” com distribuições diversas:

```
X=Normal(1, 2)  
B = Gamma(10, 0.2)
```

```
Gamma{Float64}(α=10.0, θ=0.2)
```

A função `Normal(1, 2)` cria uma “variável aleatória” com valor esperado 1 e **desvio-padrão** 2 (cuidado que essa notação é diferente da que usamos habitualmente, em que $N(a, b)$ representa uma variável aleatória com valor esperado a e **variância** b). A função `Gamma(10, 0.2)` cria uma variável aleatória binomial com parâmetros $\alpha = 10$ e $\theta = 0.2$.

Podemos achar a probabilidade de X ser menor do que 0.5 usando

```
cdf(X, 0.5)
```

0.4012936743170763

e analogamente

```
cdf(B, 0.5)
```

0.00027735209462083615

Obviamente, a probabilidade de $X \leq 1$ é 0.5:

```
cdf(X, 1)
```

0.5

A função $\text{ccdf}(x)$ é $1 - \text{cdf}(x)$, ou seja ' $\text{ccdf}(x) = \text{Prob}(\{X > x\})$:

```
ccdf(X, 0.5)
```

0.5987063256829237

```
ccdf(X, 0.5) + cdf(X, 0.5)
```

1.0

Outra função útil é $\text{quantile}(p)$, que fornece o valor de x para o qual $\text{Prob}(\{X \leq x\}) = p$, para um valor de p escolhido:

```
quantile(X, 0.4)
```

0.49330579372840055

```
quantile(B, 0.4)
```

1.7808829473194245

```
quantile(X, 0.5)
```

1.0

A função $\text{cquantile}(p)$ fornece o valor de x para o qual $\text{Prob}(\{X > x\}) = p$:

```
cquantile(X, 0.4)
```

1.5066942062715993

```
cquantile(X, 0.6)
```

0.49330579372840055

Várias outras distribuições estão disponíveis, por exemplo, [Bernoulli](#), [Binomial](#), [Poisson](#), [Exponential](#), [Chisq](#) (χ^2), etc:

Podemos gerar números aleatórios de qualquer das distribuições disponíveis usando o comando `rand`, e o nome da variável aleatória como primeiro parâmetro. Por exemplo, para criar um vetor com três amostras de uma [Gamma\(10, 0.2\)](#), use

```
b = rand(B, 3)
```

3-element Vector{Float64}:

1.1658960147438355

3.2629408691022688

1.4906257346562342

Há funções para calcular valor esperado, mediana e variância:

```
println("Valor esperado = ", mean(B))  
println("Mediana = ", median(B))  
println("Variância = ", var(B))
```

Valor esperado = 2.0

Mediana = 1.9337429229428258

Variância = 0.4000000000000001

Note que esses são os valores teóricos da média e da variância. As mesmas funções podem ser usadas para calcular os valores experimentais, a partir de um conjunto de amostras:

```
b = rand(B, 1000)  
println("Média da amostra = ", mean(b), " ≈ ", mean(B))  
println("Variância da amostra = ", var(b), " ≈ ", var(B))  
println("Mediana da amostra = ", median(b), " ≈ ", median(B))
```

Média da amostra = 1.970954877314575 ≈ 2.0

Variância da amostra = 0.3997668887890061 ≈ 0.4000000000000001

Mediana da amostra = 1.8956538669128231 ≈ 1.9337429229428258

Polinômios

O pacote [Polynomials](#) é útil para trabalhar com polinômios. Você precisa carregá-lo com `using` (e, conforme o caso, instalá-lo antes com o comando `add` no modo de pacotes).

```
using Polynomials
```

```
p=Polynomial([2, 3, 1],:x)
```

$$2 + 3 \cdot x + x^2$$

A entrada `:x` no final informa qual a variável será usada para descrever o polinômio. Preste atenção que a ordem dos coeficientes é a inversa do Matlab: o primeiro coeficiente corresponde ao termo de menor grau, e o último ao termo de maior grau.

Você pode calcular o valor do polinômio em um ponto com

```
p(3)
```

20

A função `roots` busca as raízes do polinômio (note que há um pacote `Roots`, que contém funções para calcular zeros de funções genéricas):

```
roots(p)
```

2-element Vector{Float64}:

-2.0

-1.0

Você também pode criar um polinômio a partir de suas raízes (a variável default é `x`):

```
q = fromroots([3,1])
```

$$3 - 4 \cdot x + x^2$$

e fazer operações como soma e multiplicação de polinômios normalmente

```
z = Polynomial([2, 1])
```

```
s = p * q + z
```

$$8 + 2 \cdot x - 7 \cdot x^2 - x^3 + x^4$$

```
roots(s)
```

4-element Vector{Float64}:

-1.9999999999999991

-1.1149075414767564

1.2541016883650529

2.8608058531117067

Você pode criar polinômios com outras variáveis também (repare que a sintaxe para mudar o nome da variável é diferente para `fromroots`):

```
r = Polynomial([2, 1], :z)
```

```
t = fromroots([-1, 2, 3], var = :z)
```

```
r + t
```

$$8 + 2 \cdot z - 4 \cdot z^2 + z^3$$

Operações com polinômios só funcionam se eles forem descritos em termos da mesma variável. Por exemplo, o produto `p*r` resultará em um

erro.

O pacote `Polynomials` também permite trabalhar com funções racionais (repare que deve ser usado `//`, o operador para divisão racional):

```
p = Polynomial([5, 1])
```

$5 + x$

```
q = Polynomial([5, 4, 1])
```

$5 + 4 \cdot x + x^2$

```
G = p // q
```

$(5 + x) // (5 + 4x + x^2)$

É possível calcular a função em um ponto

```
G(3)
```

0.3076923076923077

E também é possível fazer divisão polinomial usando os operadores `÷` (divisão) e `%` (resto):

```
t = q ÷ p
```

$-1.0 + 1.0 \cdot x$

```
r = q % p
```

10.0

```
t * p + r == q
```

true

Além disso, você pode calcular os resíduos de cada raiz (útil para antitransformada de Laplace e inversa da transformada z) com a função `residues`:

```
res = residues(G)
```

(Polynomial(0), Dict{ComplexF64, Vector{ComplexF64}}(-2.0 - 1.0im => [0.5 + 1.5im], -2.0 + 1.0im => [0.5 - 1.5im]))

As saídas são duas: `res[1]` é a parte *inteira* divisão do numerador pelo denominador (o resultado de $p \div q$), enquanto que `res[2]` é um dicionário, com as raízes do denominador e os resíduos correspondentes:

```
res[1]
```

0

```
p ÷ q
```

0

```
res[2]
```

Dict{ComplexF64, Vector{ComplexF64}} with 2 entries:

-2.0-1.0im => [0.5+1.5im]

-2.0+1.0im => [0.5-1.5im]

Veja como podemos recuperar a função $G(x)$ original somando os termos correspondentes a cada raiz com os resíduos correspondentes:

```
G1 = res[1]
for k in keys(res[2])
    G1 += Polynomial(res[2][k])//fromroots([k])
end
G1
```

```
(5.0 + 1.0*x) // (5.0 + 4.0*x + 1.0*x^2)
```

Filtros digitais

Várias funções úteis para processamento de sinais estão no pacote `DSP`. Você precisa carregá-lo com o comando `using` (e se necessário antes instalá-lo com o comando `add` no modo de pacotes).

```
using DSP
```

Podemos definir um filtro FIR através do vetor de coeficientes usando os comandos a seguir:

```
ωc = π/3
```

```
1.0471975511965976
```

```
N = 21
L = (N-1)/2
```

```
10.0
```

Lembre que um filtro passa-baixas ideal com corte na frequência ω_c tem resposta ao impulso

$$h_d[n] = \frac{\omega_c}{\pi} \text{sinc} \left(\frac{\omega_c}{\pi} (n - L) \right), 0 \leq n \leq N - 1.$$


Para gerar essa resposta ao impulso, criamos um vetor n com valores entre 0 e $N - 1$:

```
n = 0:N-1;
```

Para criar a resposta ao impulso, usamos a função `sinc`. Cuidado apenas com o cálculo de $n - L$: como n é um vetor, a operação $n - L$ irá gerar um erro (pois estamos tentando subtrair um escalar de um vetor).

Precisamos então usar " `.-` ", para indicar que queremos subtrair o escalar de todos os elementos do vetor:

```
hd=(wc/pi)*sinc.((wc/pi)*(n .- L))
plot(n, hd, line = :stem, marker = (:circle, 3),
      xlabel = L"n", label = L"h_d[n]")
```



Podemos calcular a resposta em frequência do filtro usando o comando `freqresp`. No pacote `DSP`, o comando `freqresp` é um pouco diferente do correspondente no `Matlab`: a entrada é uma variável do tipo `filter`, e (opcionalmente) um vetor com as frequências em que se deseja calcular a saída. Se o vetor de frequências não for fornecido, a função tem duas saídas: primeiro, a resposta em frequência desejada, e segundo, um vetor com os valores de frequência (entre 0 e π utilizados).

Vamos ver um exemplo. Primeiro definimos uma variável do tipo `filter` através da função `PolynomialRatio`, que pede como entrada dois vetores, com os coeficientes do numerador e do denominador do filtro desejado. Como o nosso filtro é FIR, o denominador é apenas 1. Cuidado porque aqui há uma pegadinha: a função `PolynomialRatio(b,a)` precisa

que tanto a quanto b sejam vetores. Se você escrever

```
hdf=PolynomialRatio(hd,1)
```

vai receber uma mensagem de erro, porque 1 é um escalar, não um vetor. Para não ter problemas, você deve escrever

```
hdf=PolynomialRatio(hd, [1])
```

como abaixo:

```
hdf=PolynomialRatio(hd, [1.0]);
```

Agora podemos definir o vetor de pontos. Para simplificar, vamos usar a função `range`:

```
 $\omega$ =range(0, $\pi$ , length=500);
```


A função `range(a, b, length=N)` retorna um vetor com N elementos. O primeiro elemento é a , e o último é b .

Agora podemos calcular a resposta em frequência do filtro:

```
Hd=freqresp(hdf,  $\omega$ );
```

Você também pode usar o comando `Hd, ω = freqresp(hdf)` - neste caso o vetor ω não precisa ser criado antes.

```
plot( $\omega/\pi$ , abs.(Hd), label = L"H_d(e^{j\omega})")  
xlabel!(L"\omega/\pi")  
ylabel!(L"|H(e^{j\omega})|")
```




Para evitar as oscilações, podemos usar uma janela de Kaiser. Suponha que desejemos uma atenuação de 60dB e uma faixa de transição de largura $\Delta\omega = \pi/5$. Então podemos calcular os parâmetros da janela de Kaiser usando a função `kaiserord()` (cuidado que a definição do parâmetro β em *Julia* é diferente da do *Matlab* por um fator de π)

```
N,α=kaiserord(π/5, 60)
```

```
(13, 1.7994885471673767)
```


Agora recalculamos a resposta ao impulso ideal e a multiplicamos pela janela de Kaiser adequada:

```
L=(N-1)/2
n=0:N-1
hd=(wc/π)*sinc((wc/π)*(n.-L))
h=hd.*kaiser(N,α)
plot(n,hd,label="Jan. retangular", line = :stem,
      marker = (:circle, 3))
plot!(n,h,label="Jan. de Kaiser",line = :stem, marker = (:square, 3))
xlabel!(L"n")
```



A nova resposta ao impulso fica

```
hf = PolynomialRatio(h, [1])
H = freqresp(hf, ω)
plot(ω/π, abs.(Hd), label="Retangular, N=21")
plot!(ω/π, abs.(H), label="Kaiser, N=13")
xlabel!(L"\omega/\pi")
ylabel!(L"|H_d(e^{j\omega})|, |H(e^{j\omega})|")
```



Atenção: o valor do parâmetro β da janela de Kaiser visto no curso deve ser dividido por π para usar no comando `kaiser` deve ser dividido por π : `kaiser(N, β/π)`

Também é possível criar filtros IIR. Por exemplo, o filtro


$$G(z) = 0.01 \frac{1 + z^{-1}}{1 - 0.8z^{-1}}$$

pode ser criado usando tanto a função `PolynomialRatio()` (cujas entradas são os coeficientes do numerador e do denominador) quanto a função `ZeroPoleGain()`, cujas entradas são os zeros, pólos e o ganho. Veja os exemplos:

```
g=PolynomialRatio(0.1*[1, 1],[1, -0.8])
```

```
PolynomialRatio{:z, Float64}(LaurentPolynomial(0.1*z-1 + 0.1),
LaurentPolynomial(-0.8*z-1 + 1.0))
```

```
G = freqresp(g, ω);
plot(ω/π, abs.(G), label = "")
xlabel!(L"\omega/\pi")
ylabel!(L"|G(e^{j\omega})|")
```




Também é possível definir um filtro através dos pólos, zeros e do ganho:

```
g1=ZeroPoleGain([-1],[0.8],0.1)
```

```
ZeroPoleGain{:z, Int64, Float64, Float64}([-1], [0.8], 0.1)
```

```
G1 = freqresp(g1, \omega);
plot(\omega/\pi, abs.(G1), label = "")
xlabel!(L"\omega/\pi")
ylabel!(L"|G_1(e^{j\omega})|")
```



Repare que se você quiser definir um filtro sem nenhum zero finito, por exemplo

$$G_2(z) = 0.2 \frac{1}{1 - 0.8z^{-1}}$$

usando a função ZeroPoleGain, você precisa usar um vetor "vazio" (com zero elementos), como

```
zvazio = Array{Float64,1}()
```


```
Float64[]
```

Veja mais detalhes sobre tipos de variáveis na seção [Tipos de variáveis](#).

```
g2 = ZeroPoleGain(zvazio, [0.8], 0.2)
```

```
ZeroPoleGain{:z, Float64, Float64, Float64}(Float64[], [0.8], 0.2)
```


```
G2=freqz(g2,\omega);
plot(\omega/\pi, abs.(G2), label = "")
xlabel!(L"\omega/\pi")
ylabel!(L"|G_2(e^{j\omega})|")
```



A saída de um filtro para uma entrada qualquer pode ser calculada usando-se a função `filt()`

```
n1 = 0:1000
x = 0.5*cos.((pi/50)n1)
y = filt(hf, x);
```

```
plot(n1, y, label = "Saída do filtro")
xlabel!(L"n")
```



Outros formatos


 PDF

 OpenOffice

```

y1 = filt(g, x)
plot(n1, y1, label = "saída do filtro IIR")
xlabel!(L"n")


```



A função `filt()` também pode ser usada definindo o filtro diretamente

pelos coeficientes do numerador e do denominador (repare que a função `filt()` aceita que a segunda entrada seja um escalar):

```
yalt = filt(h, 1, x)
plot(n1, y, label = "Calculado com hf", lw = 4)
xlabel!(L"n");
plot!(n1, yalt, label = "Calculado com h")
```



Tipos de variáveis em *Julia*

Se você estiver escrevendo um programa simples só para resolver um exercício, em geral não é importante prestar atenção se uma variável é um inteiro, um número em ponto flutuante, etc. No entanto, isso é importante quando você quer gerar código eficiente, para resolver um problema real. Também é importante para entender as mensagens de erro de *Julia*. Como a linguagem foi projetada para ser eficiente, para aplicações em cálculo numérico, em algumas situações a conversão de variáveis de um tipo para outro não é automática.

Os dois tipos de variáveis mais comuns são inteiros e "reais" (ponto flutuante). Você pode escolher entre um tipo e outro ao entrar uma variável escrevendo ou não o ponto decimal. Por exemplo,

```
n = 2
```

é um inteiro, enquanto

```
x = 2.
```

2.0

Você pode verificar o tipo de uma variável usando a função `typeof()`

```
typeof(n)
```

Int64

```
typeof(x)
```

Float64

```
q = 2//3  
typeof(q)
```

Rational{Int64}

Ao fazer uma conta contendo números de diferentes tipos, quando possível *Julia* converte todos os números para o tipo mais geral, e faz a conta considerando esse tipo.

```
n+x
```

4.0

```
n+q
```

8//3

```
x+q
```

2.6666666666666665

```
z = 1+0im
```

1 + 0im

```
z+n
```

3 + 0im

```
z+q
```

5//3 + 0//1*im

```
z+x
```

3.0 + 0.0im

Apenas inteiros podem ser usados para indexar vetores e matrizes:

```
v=[1, 2, 3].^2
```

3-element Vector{Int64}:

1
4
9

```
v[n]
```

4

```
v[x]
```

LoadError: ArgumentError: invalid index: 2.0 of type Float64

Isto pode gerar alguns erros estranhos quando você calcula o valor de um índice. Por exemplo, digamos que você queira pegar o elemento do meio de um vetor:

```
N = length(v)
```

3

```
i = N/2
```

1.5

Como o resultado não é um número inteiro, podemos arredondar o resultado para cima:

```
i = ceil(i)
```

2.0

No entanto, o resultado ainda não é do *tipo* inteiro, e você vai receber um erro se tentar acessar `v[i]`:

```
v[i]
```

LoadError: ArgumentError: invalid index: 2.0 of type Float64

Para resolver o problema, você pode converter o resultado para um inteiro: isso é fácil, usando as funções `Int()` ou `Int64()` - convertem o argumento para um inteiro de 64 bits.

```
i = Int(i)
```

2

As funções `round`, `floor` e `ceil` também podem converter o resultado automaticamente para o tipo inteiro, assim:

```
i = ceil(Int, N/2)
```

2

Também há inteiros de 128 (`Int128`), 32 (`Int32`), 16 (`Int16`) e 8 (`Int8`) bits, e inteiros sem sinal (`UInt`).

```
v[i]
```

4

```
typeof(i)
```

Int64

Você também poderia usar o operador `÷`, a divisão inteira, além da versão completa do comando `ceil`:

```
i = N÷2 + 1  
v[i]
```

4

Uma outra situação em que você pode ter um resultado inesperado é se tentar usar a função `sqrt()` com um número negativo:

```
sqrt(2)
```

1.4142135623730951

```
sqrt(-2)
```

LoadError: DomainError with -2.0:
sqrt will only return a complex result if called with a complex argument. Try sqrt(Complex(x)).

Para aumentar a eficiência da linguagem, a função `sqrt()` não testa se a entrada é positiva ou não, e escolhe se a saída é real ou complexa de acordo com o valor da entrada - na verdade existe uma versão da função

`sqrt()` para entradas reais e outra para entradas complexas:

```
sqrt(-2+0im)
```

```
0.0 + 1.4142135623730951im
```

O mesmo efeito pode ser obtido convertendo o argumento de `sqrt()` para complexo antes de chamar a função:

```
sqrt(Complex(-2))
```

```
0.0 + 1.4142135623730951im
```

Saber um pouco sobre tipos de variáveis ajuda a entender as mensagens de erro da linguagem. Por exemplo, se você definir

```
v=[1, 2]
```

```
2-element Vector{Int64}:
```

```
1  
2
```

e tentar calcular `cos(v)` (sem usar o `.` antes dos parênteses), vai receber uma mensagem de erro

```
cos(v)
```

```
LoadError: MethodError: no method matching  
cos(::Vector{Int64})
```

```
Closest candidates are:  
 cos(::T) where T::Union{Float32, Float64}  
   @ Base ./special/trig.jl:98  
 cos(::Hermitian{var"#s970", S} where {var"#s970" <: Complex, S::(AbstractMatrix{<:var"#s970"})})  
   @ LinearAlgebra ./Applications/Julia-1.9.app/Contents/Resources/julia/share/julia/stdlib/v1.9/LinearAlgebra/src/Hermitian.jl:732  
 cos(::Union{Hermitian{var"#s971", S}, Symmetric{var"#s971", S}} where {var"#s971" <: Real, S})  
   @ LinearAlgebra ./Applications/Julia-1.9.app/Contents/Resources/julia/share/julia/stdlib/v1.9/LinearAlgebra/src/Symmetric.jl:728  
 ...
```

O que essa mensagem está dizendo é que a função `cos()` recebeu como entrada um vetor, mas não está definida para entradas do tipo vetor. Para resolver o problema, use `cos.(v)`:

```
cos.(v)
```

2-element Vector{Float64}:

0.5403023058681398

-0.4161468365471424

Outra situação sutil é a diferença entre uma matriz com uma coluna e um vetor. Veja, por exemplo, as definições a seguir. Veja a descrição do tipo de um vetor ou de uma matriz:

```
a=[1, 2]
```

2-element Vector{Int64}:

1

2

```
a1 = [1 2]
```

1×2 Matrix{Int64}:

1 2

Para definir um vetor coluna, você pode separar os elementos com `,` ou `;`. No entanto, no caso de matrizes é necessário usar `;`.

```
B=[1 2;3 4]
```

2×2 Matrix{Int64}:

1 2

3 4

Repare que `a` é `1×2 Matrix{Int64}`, ou seja: uma lista unidimensional de números inteiros, enquanto que `B` é `2×2 Matrix{Int64}`, ou seja: um arranjo bidimensional de números inteiros. Em outras palavras, `a` é um vetor e `B` é uma matriz.

A transposta de `a` é também um vetor-linha:

```
a'
```

1×2 adjoint(::Vector{Int64}) with eltype Int64:

1 2

Você pode criar um vetor com `randn` usando um único argumento:

```
x=randn(3)
```

3-element Vector{Float64}:

-0.046899537026286416

0.7136693922329185

1.5275488873915875

No entanto, se fizer como no *Matlab*, e usar dois argumentos, o que você obtém sempre é considerado uma matriz:

```
y=randn(3,1)
```

```
3×1 Matrix{Float64}:  
 0.2019849284909708  
-0.3037616751026521  
 0.6677422173621351
```

Veja que a descrição mudou: no primeiro caso, x é um `3-element Vector{Float64}`, que é equivalente a um `Array{Float64,1}`, enquanto que no segundo caso, y é um `3×1 Matrix{Float64}`, que é equivalente a um `Array{Float64,2}`. x é vetor, enquanto que y é uma matriz que acontece de ter uma única coluna.

Na maior parte das vezes você não vai ter problema nenhum com isso, nem vai reparar na diferença. Você pode ter problemas no entanto com algumas operações, vamos ver uns exemplos a seguir e como consertar. Imagine que você queira calcular os produtos escalares de x por x , x por y , y por y . Como vimos antes, você pode usar tanto a expressão $x'*y$ quanto $x \cdot y$ (o operador \cdot é obtido digitando-se `\cdot[TAB]`):

```
x'*x
```

```
2.8449291713546603
```

```
x'*y
```

```
1×1 adjoint(::Vector{Float64}) with eltype Float64:  
 0.7937504715092442
```

```
y'*y
```

```
1×1 Matrix{Float64}:  
 0.5789487354463727
```

Repare que cada caso resultou numa variável de tipo diferente: $x' * x$ resultou em um escalar, $x' * y$ resultou em um "vetor"-linha com um único elemento, e $y' * y$ resultou em uma "matriz" 1x1.

Novamente, em geral isso não vai causar muito problema, a não ser que você queira usar o resultado da operação como entrada de uma função que só aceita escalares:

```
exp(x'*x)
```

```
17.200340348973672
```

```
exp(x'*y)
```

```
1×1 Matrix{Float64}:  
 0.5789487354463727
```

```
2.2116757176947766
```

Isso em geral não causa problemas:

```
exp.(x'*y)
```

```
1×1 Matrix{Float64}:
```

```
2.2116757176947766
```

Uma forma mais elegante de fazer a mesma coisa (porque usa melhor a memória, já que a saída é realmente um escalar) é usar o operador `·` (`\cdot` [TAB]):

```
x·x
```

```
2.8449291713546603
```

```
x·y
```

```
0.793750471509244
```

```
exp(x·x)
```

```
17.200340348973672
```

```
exp(x·y)
```

```
2.2116757176947766
```

Desta vez o resultado sempre foi um escalar, como desejado!

Julia pode criar código eficiente, mas para isso é necessário tomar alguns cuidados. Se você tiver curiosidade de entender melhor como usar a linguagem bem, veja [esta página](#) e [esta outra](#).